Sensors and Materials, Vol. 10, No. 7 (1998) 435–460
MYU Tokyo

S & M 0345

Circuit Modeling and SPICE Simulation of Mixed-Signal Microsystems

Hoan H. Pham and Arokia Nathan

Department of Electrical and Computer Engineering
University of Waterloo, Waterloo, Ontario, Canada N2L 3G1

(Received March 18, 1997; accepted January 11, 1998)

Key words: mixed-signal circuit simulation, constant voltage and constant charge operation modes, electrostatic pull-in

Design of microsystems in a circuit simulation environment requires an equivalent circuit representation that not only accommodates mixed signals from very different energy domains but also their strong coupling, which in many cases can turn out to be highly nonlinear. Here, the most important step for expedient yet accurate system-level simulations lies in the description of underlying coupling elements and associated variables in terms of compact equivalent circuits. In this paper, we present a technique to describe coupling behavior using dependent passive elements which is very effective in synthesis of simple and compact equivalent circuits for simulation of electromechanical and electrothermal microsystems. A detailed description of the circuit synthesis procedure is given along with three SPICE simulation examples comprising an electrothermal resistor, electrostatically actuated parallel plate capacitor and torsional micromirror. In the last two examples, we examine device operation from the viewpoint of constant voltage and constant charge operation modes, and their effect on electrostatic pull-in.

1. Introduction

The design of successful microsystems relies on effectively accommodating the interaction of mixed-signal types at the component- and system-level in a self-consistent manner. Here, the coupling of mixed signals often manifests itself in some elements whose values are dependent on the signals within their energy domain or from other energy domains. These elements are called coupling elements; in bond graph terminology,[1] these

are referred to as multiport elements. Description of coupling elements and associated variables in terms of an equivalent circuit is an important step in the circuit representation of the entire microsystem. The motivation underlying equivalent circuit approaches for microsystem design stems from the need to account for interactions between the microtransducer and co-integrated support circuitry; the latter is necessary for improved accuracy, reliability, and functionality of the microsystem. Additionally, this opens new prospects for use of powerful circuit simulators (*e.g.*, SPICE[2]) for low-cost system-level design verification, design optimization, and sensitivity analysis. The difficulty here lies in accommodating simplicity, comprehensibility, insight, and effectiveness in handling nonlinear coupling.

This paper, which constitutes a comprehensive version of a previous note,[3] details the equivalent circuit synthesis procedure to handle coupling elements in electrothermal and electromechanical microsystems using dependent passive circuit elements. Section 2 briefly describes the various approaches that have been employed for simulation of mixed-signal microsystems. In section 3, we describe our synthesis approach, which is based on use of multivariate polynomial dependent sources. Section 4 illustrates a SPICE simulation example of an electro-thermal resistor, and sections 5 and 6 describe the circuit modeling and simulation of two electrostatically actuated microsystems: a vertically actuated parallel-plate capacitor and the torsional micromirror. Here we introduce the notion of constant charge operation and investigate electrostatic pull-in behavior from the viewpoint of constant voltage and constant charge operation modes.

2. Simulation Approaches

Important goals in circuit synthesis for mixed-signal systems are that the equivalent circuit be simple, compatible with standard circuit simulators, comprehensible for gaining preliminary insight into system behavior under different geometric and operating conditions, and effective in handling nonlinear coupling. The simpler the circuit, the better it is in terms of realization, verification, and understanding. The model equations governing coupling may have several different equivalent forms, and one form may lead to better circuit realization than the others. Thus, choosing the appropriate form is crucial particularly from the standpoint of handling nonlinear coupling. There are several approaches for simulation of mixed-signal microsystems; each varies in terms of the degree of simplicity and effectiveness.

• Small signal model:[4] For small deviations from some equilibrium point, the nonlinear dependence may be linearized. However, this restricts simulation of system behavior to a narrow range of parameter values.

• Decoupling:[5] Each signal is treated in its native energy domain, and the coupling is realized via exchange of their values. The disadvantage here lies in the large number of netlist regenerations and simulation runs. The latter can turn out to be exorbitant particularly for tightly coupled problems, where iterations may converge very slowly or even diverge.

• Analog computer techniques:[6,7] Here, the governing differential equations are solved

using basic computing elements (circuits) such as integrators, adders and multipliers. The main problem with this approach is that the circuit obtained may be complex and may contain an excessive number of circuit elements, providing only limited insight into system behavior. For example, solution of a second-order linear differential equation requires two integrators. However, in most cases, an RLC circuit suffices. More importantly, the roles of R, L, and C in the circuit give valuable information about system behavior.

• Dependent passive elements:[3] Since the nonlinear coupling manifests itself in coupling elements, equivalent circuits for these elements can be synthesized through use of multivariate polynomial dependent sources that accurately describe the interaction of the mixed signal and associated non-linearity. This is the approach that we adopt for simulation of mixed-signal systems. The background for this approach stems from the theory of bond graphs. Here, the various types of coupling elements can be described in terms of dependent passive elements such as dependent resistors (R), capacitors (C), and inductors (L). The method for synthesis of equivalent circuits for dependent passive elements is described in the following section.

3. Circuit Synthesis for Coupling Elements

In many mixed-signal systems, passive elements such as R, C and L may be dependent on some variable (signal) x. These passive elements serve as a source of coupling. Some circuit simulators do not support dynamically varying values of passive elements during the course of simulation. To deal with this problem, we need to express the dependence in another form. Fortunately, multivariate polynomial dependent sources have become a standard feature in state-of-the-art SPICE-like simulators. The trick here is to use dependent sources instead. To facilitate discussion, let us denote a dependent passive element by $P \in \{R, C, L\}$. Its value $P(x)$ is dependent on some signal x such as temperature or displacement. We obtain an approximate or exact, if possible, polynomial form for either $P(x)$ or $\dfrac{1}{P(x)}$

$$P(x) = P_o[1 + f(x)] \tag{1}$$

$$\frac{1}{P(x)} = \frac{1}{P_o}[1 + g(x)], \tag{2}$$

where $f(x)$ and $g(x)$ are some polynomials of the external signal x. Note that with the above forms, either $P(x)$ or $\dfrac{1}{P(x)}$ can be decomposed into two components, one of which is independent of x while the other can be realized using a polynomial dependent source. Either of the polynomial forms in eqs. (1) or (2) will suffice in describing the coupling element.

3.1 *The dependent resistor*

In this case, P in eqs. (1) and (2) denotes a resistor, $R \equiv R(x)$.

• With the polynomial form in eq. (1), by choosing the constitutive relation $V = iR$, we have $V = iR_o[1 + f(x)] = iR_o + iR_of(x)$. We put an independent resistor R_o in serial connection. Let V_o denote the voltage across the resistor R_o. We have $V = V_o + V_of(x)$. The dependence of R on x can now be described by an independent resistor R_o and a dependent voltage-controlled voltage source (VCVS) $V_of(x)$ (see Fig. 1(a)).

• With polynomial form in eq. (2), by choosing the constitutive relation $i = \frac{V}{R}$, we have $i = \frac{V[1 + g(x)]}{R_o}$. Here, i is the current through the resistor $R(x)$, and V is the voltage across it. This current can also be viewed as the current through the resistor R_o, the voltage across which is $V[1 + g(x)]$. This suggests a method to realize $R(x)$ using a dependent source. We put into a serial connection an independent resistor R_o and a VCVS of value $-Vf(x)$ (see Fig. 1(b)). Note the minus sign for the VCVS. The voltage across the independent resistor R_o is $V - (-Vg(x)) = V[1 + g(x)]$, as expected. The dependence of R on x in this case can also be described by an independent resistor (i.e., R_o) and a VCVS (i.e., $-Vg(x)$).

Note that from the constitutive relation $i = \frac{V}{R} = \frac{V}{R_o}[1 + g(x)] = \frac{V}{R_o} + \frac{V}{R_o}g(x)$, the current i can also be viewed as the sum of two currents $\frac{V}{R_o}$ and $\frac{V}{R_o}g(x)$. This suggests an alternative way to realize $R(x)$. We put in parallel connection the independent resistor R_o. Let i_o denote the current through the resistor R_o. We have $i = i_o + i_og(x)$. The dependence of R on x can now be described by a dependent current-controlled current source (CCCS) of value $i_og(x)$ and an independent resistor R_o (see Fig. 1(c)). In some circuit simulators, the controlling current in a CCCS must be the current through a voltage source; in this case, we may re-express the relation as $i_o = \frac{V}{R_o}$ and $i = i_o + \frac{V}{R_o}g(x)$; and the CCCS now becomes a voltage-controlled current source (VCCS).

Figures 1(a) and 1(b) show circuit realization of a dependent resistor. The choice of constitutive relation $V = \phi(i)$ or $i = \phi^{-1}(V)$ (where ϕ is some function), in this case, is determined by the availability of the polynomial forms (1) or (2). An example of the dependent resistor as a coupling element is the electro-thermal resistor, which is common to most devices in the thermal microtransducer family. Model equations, circuit synthesis using the above method, and SPICE simulation of the electro-thermal resistor are presented in section 4 and Appendix A.

3.2 *The dependent capacitor*

In this case, P is a capacitor $C \equiv C(x)$. Here, we replace i by q and R_o by $\frac{1}{C_o}$ ($V = iR \leftrightarrow V = q\frac{1}{C}$.

• With the polynomial form in eq. (1), by choosing the constitutive relation $q = CV =$



(a) $\mathbf{R(x) = R_o[1 + f(x)]}$

(b) $\mathbf{R(x) = \frac{1}{R_o}[1 + g(x)]}$

(c) $\mathbf{R(x) = \frac{1}{R_o}[1 + g(x)]}$

Fig. 1. Transformation of the coupling resistor to a dependent source.

$C_o[1 + f(x)]V$, we can view q as the charge on the plate of the capacitor C_o, the voltage across which is $V[1 + f(x)]$. This is similar to the case of the dependent resistor described by the polynomial form in eq. (2). A realization of $C(x)$ is shown in Fig. 2(a). The circuit here is similar to that shown in Fig. 1(b). Note that although q can be expressed as $q = C_oV + C_oVf(x) = q_o + q_of(x)$, where $q_o = C_oV$, the circuit for $C(x)$ can turn out to be complex. The current i through C is $i = \dot{q} = \dot{q}_o + \dot{q}_o f(x) + q_o \dot{f}(x)$, which needs three currents in parallel connection: a current i_o through the independent capacitor C_o, a CCCS of value $i_of(x)$, and a VCVS of value $C_oV\dot{f}(x)$. Obviously, such a realization is inferior to the realization shown in Fig. 2(a).



(a) $C(x) = C_o[1 + f(x)]$



(b) $\frac{1}{C(x)} = \frac{1}{C_o}[1 + g(x)]$

Fig. 2. Transformation of the coupling capacitor to a dependent source.

• With the polynomial form in eq. (2), by choosing the constitutive relation $V = \frac{q}{C}$, we obtain $V = \frac{q}{C_o}[1 + g(x)] = V_o + V_o g(x)$, where $V_o = \frac{q}{C_o}$. The voltage V across the dependent capacitor C can be viewed as the sum of two voltages: the voltage V_o of the independent C_o and the VCVS of value $V_o g(x)$. A realization of $C(x)$ in this case is shown in Fig. 2(b). The circuit has a structure similar to that shown in Fig. 1(a).

Again, as with the dependent resistor, the choice of constitutive relation $q = \phi(V)$ or $V = \phi^{-1}(q)$ is determined by the availability of the polynomial forms in eqs. (1) or (2).

We now have a method to synthesize equivalent circuits for dependent passive elements whose value can be described either by forms in eqs. (1) or (2). The question arises as to which form should be used. The answer to the question depends on the nature of $P(x)$; as a rule of thumb, if the value of $P(x)$ can become infinitely large, then the form $\frac{1}{P(x)}$ is preferred.

In some circuit simulators, passive elements can be described as a function of nodal voltages. This feature provides another means for describing coupling and circuit synthesis. One may then argue about the merit of the above method of describing dependent

passive elements in terms of VCVSs. However, as it will be shown later, the method has some advantages. First, not all circuit simulators can describe passive elements as functions of nodal voltages. Second, with the above equivalent circuits, we can obtain, without any extra cost, the current i through the dependent resistor, which is $\frac{V_o}{R_o}$ (Figs. 1(a) and 1(b)) or the charge q on the dependent capacitor, which is C_oV_o (Figs. 2(a) and 2(b)). Otherwise, we may need extra circuits to compute these values; indeed, such a saving can become significant when dealing with synthesis of equivalent circuits for distributed-parameter modeling.

To illustrate the above method of circuit realization for dependent passive elements, we present three examples. The first example (section 4) deals with the electrothermal resistor. The second example (section 5) deals with a vertically actuated parallel-plate capacitor, and the third example (section 6) deals with a torsional micromirror, which is used in light modulation for imaging applications.[8,9]

4. Electrothermal Resistor

An electro-thermal resistor can be viewed as (i) an electrical element with electrical resistance R, which provides Joule heat $P = Vi = i^2R$ and (ii) a thermal object with thermal resistance R_{th}, which accounts for heat exchange with the environment of temperature T_∞ and with thermal capacitance C for storage of heat energy. A current through an electro-thermal resistor generates Joule heat; the latter is a function of the electrical resistance R. This in turn heats the resistor, thus changing its electrical resistance. This clearly illustrates the nature of the coupling between the electrical and thermal domains.

4.1 *Model equations*

For temperature variations that are not too large, the electrical resistance R can be modeled in terms of its temperature coefficient α:

$$R(T) = R_o[1 + \alpha T]. \tag{3}$$

Assume that the resistor is isothermal and that heat convection can be described using a heat transfer coefficient h. Then the model equations for the electrothermal system read:

$$V = iR_o[1 + \alpha T] \quad \text{electrical domain} \tag{4}$$

$$Vi = C\frac{dT}{dt} + h[T - T_\infty] \quad \text{thermal domain.} \tag{5}$$

Here, V is the voltage across the resistor and i is the current through it. Equation (5) states that the supplied electrical power Vi is partly channeled to the heat capacitor (C) for heat energy storage and is partly channeled to the thermal resistor ($R_{th} = 1/h$) for heat exchange with the environment.

4.2 *Circuit synthesis*

The temperature dependence of the electrical resistor *R(T)* (eq. (3)) is of the exact polynomial form given in eq. (1), where $f(x) = f(T) = \alpha T$. Therefore, the dependent resistor *R(T)* can be realized using the circuit structure in Fig. 1(a). Let V_{12}, instead of V_o, denote the voltage across the resistor R_o placed into a serial connection. Here, the subscripts (1, 2) denote the nodes in the circuit. The current i through the resistor *R(T)* is identical to the current through the independent resistor R_o. Therefore,

$$i = \frac{V_{12}}{R_o}. \tag{6}$$

The Joule heat *Vi* is:

$$Vi = \frac{VV_{12}}{R_o}, \tag{7}$$

which can be realized using a bivariate polynomial VCCS of value $\frac{VV_{12}}{R_o}$. The model equation for the entire system can now be expressed as:

$$V = V_{12} + V_{12}\alpha T \tag{8}$$

$$V\frac{V_{12}}{R_o} = C\frac{dT}{dt} + h(T - T_\infty). \tag{9}$$

Figure 3 shows the resulting equivalent circuit for the electrothermal system. Here, the temperature T is given by the voltage V_3. Figure 4 shows the SPICE simulation of temperature T as a function of the input voltage V. The SPICE code for the above simulation is presented in Appendix A.



Fig. 3. Equivalent circuit for electrothermal system.



Fig. 4. SPICE simulation of temperature as a function of applied voltage in electrothermal system.

5. Vertically Actuated Parallel-Plate Capacitor

Figure 5 shows a vertically actuated parallel-plate capacitor constrained by a spring with a spring coefficient k. The friction is modeled as a damper b. Here, F denotes the electrostatic force acting on the moving plate, x the plate vertical displacement, q the charge on the plate, and V the voltage across the capacitor. When a voltage V is applied across the plate, a charge q is induced, creating an electrostatic force causing a displacement x, which in turn changes the capacitance. Hence, the capacitor can be considered as a coupling element. In bond graph terminology, it is a two-port C-field. In modeling the electromechanical system, a constitutive relationship between the variables F, x, V, and q is found by expressing C in terms of x and F in terms of V and x.

5.1 *Model equations*

The equations for electrical and mechanical operation are:

$$V_p = V_{in} - iR \qquad (i = \dot{q}) \tag{10}$$

$$F = m\ddot{x} + b\dot{x} + kx. \tag{11}$$



Fig. 5. Vertically actuated parallel-plate capacitor. Here, V_{in} = external voltage, R = resistor, V_p = voltage across capacitor plates, d = initial gap, m = moving-plate mass, k = spring constant, b = damper, x = plate vertical displacement, q = charge on the moving plate, and F = electrostatic force.

For the capacitance C, we employ

$$C = \frac{\varepsilon A}{d - x} \tag{12}$$

$$= \frac{\varepsilon A}{d}\frac{d}{d-x} = C_o \frac{d}{d-x}, \tag{13}$$

where C_o is the initial capacitance (when $x = 0$), d the initial gap between the two plates, and ε the permittivity of the dielectric between the plate, which is assumed to be air.

For the force F, we employ

$$F = \frac{\varepsilon A}{2(d-x)^2} V_p^2. \tag{14}$$

The entire system can be modeled using eqs. (10), (11), (12) and (14).

5.2 *Difficulties*

A difficulty arises due to the nonlinear dependence of eqs. (12) and (14). Using this form of capacitance might require Taylor expansion,[10] small signal approximation,[4] or a time-varying capacitor model,[11] since

$$\lim_{x \to d} C = \lim_{x \to d} \frac{\varepsilon A}{d - x} = \infty, \tag{15}$$

which implies that when the displacement x approaches d, the capacitance C becomes infinitely large. To deal with this problem, we explore an alternative form of the constitutive relation:

$$\frac{1}{C} = \frac{1}{C_o}\left(1 - \frac{x}{d}\right), \tag{16}$$

which is of the form in eq. (2), in which $g(x) = \frac{-x}{d}$. It is noted that although C given by eq. (12) does not have an exact polynomial form, $\frac{1}{C}$ does, allowing use of the multivariate polynomial dependent source described in section 3. Furthermore, by replacing V in eq. (14) with $q = VC$, we can express the force F in terms of charge in a simpler form:

$$F = \frac{q^2}{2C_o d}. \tag{17}$$

Once q is known, this form is more suitable for the circuit synthesis of F.

5.3 *Circuit synthesis*

From eq. (16), C can be realized using the circuit structure in Fig. 2(a). Let V_{23}, instead of V_o, denote the voltage across the capacitor C_o placed into a serial connection. As mentioned earlier, the charge $q = \int i dt$ on the capacitor C is also the charge q on the capacitor C_o:

$$q = C_o V_{23}. \tag{18}$$

The voltage V_p across the dependent capacitor and the force F (eq. (17)) can be expressed as:

$$V_p = V_{23} - V_{23}\frac{x}{d} \tag{19}$$

$$F = \frac{C_o}{2d} V_{23}^2. \tag{20}$$

Thanks to the capacitor C_o, we have also managed to eliminate the need to integrate $\dot{q}$. In fact, q is eliminated and replaced by $C_o V_{23}$ (eq. (18)). Equation (20) suggests the realization of the force F using a polynomial VCVS of value $\frac{C_o}{2d} V_{23}^2$. Our next step is to obtain x without using an integrator. Employing the analogy described in Table 1, we can realize eq. (11) using a VCVS and a serial RLC circuit in which kx is the voltage across the capacitor, whose capacitance is $\frac{1}{k}$. This voltage can be denoted by V_6. We obtain $x = \frac{V_6}{k}$ and eq. (19) becomes

$$V_p = V_{23} - V_{23}\frac{V_6}{kd}. \tag{21}$$

The expression in eq. (20) for F is better suited for SPICE simulation than that in eq. (14). The model equation for the entire electromechanical system is now (10), (11), (20) and (21). Figure 6 shows the equivalent circuit for the vertically-actuated parallel plate capacitor. Here, F is given by the voltage V_4 and V_p by the voltage V_2.

5.4 *Electrostatic pull-in and constant charge operation*

One of the problems related to the operation of a vertically actuated parallel-plate capacitor is electrostatic pull-in. At a given voltage, the electrostatic force increases proportionally to the charge on the plate, which, in turn, is proportional to the capacitance. The capacitance can become infinitely large when the moving plate approaches the fixed plate (eq. (15)). In contrast, the reactive spring force is kx, the maximum value of which is only kd. As a result, pull-in occurs when the gap approaches d. Quantitatively, pull-in occurs when $x \geq x_c = \frac{d}{3}$,[12] which holds true regardless of the value of spring constant k. Here, x_c is the critical gap distance for pull-in and the voltage v_c associated with bringing

Table 1
Analogy between the electrical and (translational) mechanical domains.

Electrical	Translational mechanics
Voltage V	Force F
Charge q	Displacement x
Current i	Velocity $\dot{x}$
$\frac{di}{dt}$	Acceleration $\ddot{x}$



Fig. 6. Equivalent circuit for the vertically actuated parallel-plate capacitor in Fig. 5.

the plate to x_c is called the critical collapse voltage. There are several approaches to deal with the problem of electrostatic pull-in. The simplest way is to avoid displacements larger than x_c. Alternatively, the fixed plate can be covered with an insulator to act as a landing base; here, pull-in still occurs, but this time on the landing base rather than on the fixed plate. From a simulation standpoint, pull-in may cause divergence, bringing the simulation to an abrupt stop. What if we wish to have a full control of plate movement? Let's take another look at eq. (17). Suppose that we can keep a constant charge $q = q_o$ on the plate. We call this *constant charge operation*, as opposed to the traditional *constant voltage operation*, in which constant voltage is always applied across that capacitor. The former operation mode gives rise to an electrostatic force that is constant, independent of x, thus delaying or even eliminating the onset of electrostatic pull-in. Without going into the details of constant charge operation, the simplest method is to charge the plate quickly to q_o and then cut the connection. The charging process, however, should be much faster than the mechanical response. In a circuit simulation, a constant charge on the plate can be simulated by using a constant voltage source across the capacitor C_o of Fig. 6 (eq. (18)). Here, the displacement x can assume any value from 0 to d. To obtain larger x, we need a larger charge q_o or a larger initial voltage $V_o = \frac{q_o}{C_o}$. Taking advantage of the behavior of C in eq. (15), and using a clever pulse sequencing scheme, we can obtain sufficiently large charge using a much smaller voltage.

Figure 7 shows displacement as a function of plate voltage for constant charge and constant voltage operation modes obtained from SPICE dc-simulation using the equivalent circuit in Fig. 6. The data for plate geometry and spring constant is taken from ref. 12. For constant voltage operation, divergence occurs when x approaches $x_c = \frac{1}{3}d$, which is in very good agreement with the graph shown in ref. 12. For constant charge, the plate can have any displacement between 0 and d, and divergence occurs only when x approaches d. The SPICE-code for the simulation is presented in Appendices B and C.

6. Torsional Electrostatic Micromirror

Figure 8 shows a torsional electrostatic mirror which consists of a rotating plate *RP*, two fixed electrode plates EP_1 and EP_2, and two landing plates LP_1 and LP_2. The rotating plate *RP* has length $2l_o$, rotates about the pivot point O in the center, and is grounded. The fixed electrode plates EP_1 and EP_2 have length l, with $l < \frac{1}{2}l_o$. When one of the electrodes is biased, the *RP* rotates under the influence of the electrostatic force. If the bias voltage is large enough, electrostatic pull-in occurs and the *RP* contacts the closest *LP*.

In many aspects, modeling the torsional mirror is similar to the vertically actuated parallel plate capacitor except that we have rotational instead of translational movement (see Table 2). The difference, however, is that an expression for the capacitance is not readily available. In what follows, we develop model equations for capacitance C and



Fig. 7. SPICE simulation of displacement as a function of applied voltage: constant voltage (solid curve) vs constant charge (dashed curve) operation modes.



Fig. 8. Schematic of torsional micromirror. Here, d = initial gap, $2l_o$ = rotating mirror length, l = fixed electrode length, and α = angle of rotation.

Table 2
Dual pairs: translational mechanics vs rotational mechanics.

Translational mechanics	Rotational mechanics
Force F	Torque τ
Displacement x	Relation angle α
Mass m	Moment of inertia I
Spring constant k	Torsion constant κ
Damper b	Damper b

torque τ using both analytical approximation and numerical simulation with the boundary element method (the latter for verification), and we investigate the electrostatic pull-in under constant charge and constant voltage operation modes.

6.1 *Capacitance and torque*

There are three main mutual capacitors, C_{01}, C_{02}, and C_{12}, for the pair of plates (*RP*, EP_1), (*RP*, EP_2), and (EP_1, EP_2), respectively. Due to symmetry, we have $C_{01} = C_{02}$. Furthermore, $C_{12} << C_{01}$, and therefore we can ignore the role of C_{12}. When one of the fixed electrodes, say EP_1, is biased and the other (EP_2) grounded, the electrostatic interaction is mainly from the biased EP_1 and the left half of *RP*. In this case, we are interested in the capacitor $C \equiv C_{01}$ (Fig. 9). The capacitance can be approximated as follows. By slicing the capacitor C_1D_1CD into small parallel-plate capacitors, we can approximate the capacitance as the sum of the small capacitances. The dashed region in Fig. 9 shows a small parallel-plate capacitor element at a position x from the intersection point I; its length is dx and its gap is $\tan(\alpha)x \approx \alpha x$. Note that it occurs in the case where $d << l$; therefore the approximation $\tan(\alpha) \approx \sin(\alpha) \approx \alpha$ is justified, viz.,

$$C \approx \varepsilon W \int_{IC}^{ID} \frac{dx}{\alpha x} \tag{22}$$

$$\approx \varepsilon W \frac{\ln \dfrac{ID}{IC}}{\alpha} \tag{23}$$

$$\approx \varepsilon W \frac{\ln\left(1 - \dfrac{l}{ID}\right)}{-\alpha} \tag{24}$$

$$\approx \varepsilon W \frac{\ln\left(1 - \dfrac{l}{IO_1}\right)}{-\alpha} \tag{25}$$

$$\approx \varepsilon W \frac{\ln\left(1 - \dfrac{l}{d}\alpha\right)}{-\alpha}, \tag{26}$$

where W is the width of the plates. In going from eq. (23) to eq. (24), we use $IC = ID - l$; from eq. (24) to eq. (25), we approximate ID as IO_1 by assuming that $ID >> DO_1$, and from eq. (25) to eq. (26), we approximate IO_1 as $\dfrac{d}{\tan(\alpha)} \approx \dfrac{d}{\alpha}$. By letting $x = -\dfrac{1}{d}\alpha$, we can rewrite eq. (26) as:



Fig. 9. Capacitor formed by plates RP and EP_1 in Fig. 8. Here, $O1$ = projection of pivot O on fixed electrode plane, AB = mirror plane, CD = left fixed electrode, C_1D_1 = projection of CD on mirror plane.

$$C = C_o \frac{\ln\left(1 - \frac{l}{d}\alpha\right)}{-\frac{l}{d}\alpha} \tag{27}$$

$$= C_o \frac{\ln(1+x)}{x}, \tag{28}$$

where C_o is the capacitance when $\alpha = 0$. Equation (28) is still not quite suitable for simulation because it has an indeterminate form at $x = 0$. We need to resort to using Taylor-series expansion. There are two choices: either we expand (i) $\frac{\ln(1+x)}{x}$ or (ii) $\frac{x}{\ln(1+x)}$. The latter is preferable because it is easier to deal with 0 than with ∞ when x approaches -1. Therefore, we work on the $\frac{1}{C}$ form:

$$\frac{1}{C} \approx \frac{1}{C_o} g(x), \text{ with } g(x) = 1 + \frac{1}{2}x - \frac{1}{12}x^2 + \frac{1}{24}x^3 - \frac{19}{720}x^4 + \text{K } O(x^{(n+1)}) \tag{29}$$

With the model equation for the capacitance C, we now obtain model equations for the torque τ, under two operation modes:

$$\tau = \frac{1}{2}V^2 \frac{dC}{d\alpha} \approx \begin{cases} \frac{1}{2C_o}\frac{l}{d}q^2 g'(x) & \text{constant charge} \\ \frac{1}{2}C_o \frac{l}{d}V^2 \frac{g'(x)}{g^2(x)} & \text{constant voltage.} \end{cases} \tag{30}$$

To evaluate the validity of the above approximations for capacitance C and torque τ, we employ numerical simulation for verification using the panel method.[13]

6.2 *Verification using the panel method*

In the panel method, the planar surfaces of the mirror, fixed and landing electrodes are meshed into small panels. On each panel, the charge distribution is assumed constant. For

each pair of panels i and j, the contribution of the charge on panel j to the potential at panel i is collected as a coefficient p_{ij} of a large and dense matrix P. The vector charge q on the panels and the vector potential ϕ at the centers of the panels can be expressed via the matrix P as:

$$\phi = Pq \, . \tag{31}$$

If the potential ϕ on the planar surfaces is specified, we can solve eq. (31) for the charge distribution q. Note that if an iterative method is used to solve eq. (31), the matrix-vector multiplication can be accelerated using multipole[14] or exponential expansion,[15,16] which also requires no explicit storage for the matrix P. From the solution of eq. (31), we obtain the charge and torque profiles for each panel on the mirror and the associated capacitance and torque acting on the mirror.

Figures 10(a) and 10(b) show typical charge and torque profiles on the mirror. The data for the mirror geometry is taken from ref. 8, in which $2l_o = 40$ μm, $W = 30$ μm, $d = 1.4$ μm, and l is chosen to be 18 μm. We can now compare the analytical approximation of C and τ with the numerical simulation result.

6.3 *Analytical approximation vs numerical simulation*

Figure 11 shows the capacitance C as a function of the angle of rotation α obtained from using numerical simulation, the logarithm form, and a third-order polynomial approximation (eq. (29)). Since the third-order approximation gives good agreement with the other two forms, it is chosen as the model equation for capacitance in the SPICE-simulation.

Figure 12 shows the torque τ as a function of the angle of rotation α also obtained from using numerical simulation, the logarithm form, a third-order and a seventh-order polynomial approximation. In contrast to the capacitance, the third-order approximation for the torque does not yield good agreement with the other forms. However, a seventh-order approximation yields good agreement with the simulation result. Thus, it is chosen as the model equation for torque in the SPICE simulation.

6.4 *SPICE simulation*

The model equations used in the SPICE simulation for capacitance and torque are summarized in Table 3. Figure 13 shows the angle of rotation as a function of the applied voltage obtained from the SPICE simulation. The data is in good agreement with values presented in ref. 8. For the SPICE simulations, we need to determine another important parameter: the torsion constant κ. In the next section, we will describe a method to extract κ from the values of the measured critical pull-in voltage.

6.5 *Critical angles and critical voltages*

To determine the critical angle and critical voltage associated with pull-in, we need the following system of equations for α:

$$\begin{aligned} \tau(\alpha) &= \kappa\alpha \\ \frac{\partial\tau(\alpha)}{\partial\alpha} &= \kappa, \end{aligned} \tag{32}$$



(a)



(b)

Fig. 10. (a) Charge and (b) torque profiles.

where τ is given in eq. (30). The system in eqs. (32) can be solved by first eliminating κ to yield:

$$\alpha \frac{\partial \tau(\alpha)}{\partial \alpha} - \tau(\alpha) = 0. \tag{33}$$



Fig. 11. Comparison of different approximations for the capacitances: solid line = polynomial form (n = 3); + = logarithm form; and o = numerical simulation.



Fig. 12. Comparison of different approximations for the torque: solid line = polynomial form (n = 7); dashed line = polynomial form (n = 3); + = logarithm form; and o = numerical simulation.

Table 3
Summary of model equations for C and τ used in SPICE simulations.

Capacitance	$\frac{1}{C(\alpha)} = \frac{1}{C_o}\left(1 - \frac{1}{2}\frac{l}{d}\alpha - \frac{1}{12}\left(\frac{l}{d}\right)^2\alpha^2 - 2\times\frac{1}{24}\left(\frac{l}{d}\right)^3\alpha^3\right)$
Torque	$\tau(\alpha) = \frac{1}{2C_o}\frac{l}{d}q^2\left(\frac{1}{2} + \frac{1}{6}\frac{l}{d}\alpha + \frac{1}{8}\left(\frac{l}{d}\right)^2\alpha^2 + \mathrm{K} + 2\times\frac{33953}{453600}\left(\frac{l}{d}\right)^7\alpha^7\right)$



Fig. 13. SPICE simulation of rotation angle as a function of applied voltage.

Expressing α in terms of $x = -\frac{l}{d}\alpha$ and using eq. (30), we can rewrite eq. (33) as:

$$\begin{cases} xg''(x) - g'(x) = 0 & \text{constant charge} \\ x(g''(x)g(x) - 2(g'(x))^2) - g'(x)g(x) = 0 & \text{constant voltage.} \end{cases} \tag{34}$$

Solving eq. (34) for x_q and x_v for constant charge and constant voltage operation modes, we obtain:

$$x_q = -0.71 \tag{35}$$

$$x_v = -0.44. \tag{36}$$

The respective critical angles α_q and α_v for constant charge and constant voltage operation modes are then

$$\alpha_q = 0.71\frac{d}{l} \tag{37}$$

$$\alpha_v = 0.44\frac{d}{l}. \tag{38}$$

Substituting the above values into eq. (30) and using the first equation of system (32), we obtain the following corresponding critical voltages:

$$v_{c,q} = 1.34\frac{d}{l}\sqrt{\frac{\kappa}{C_o}} \tag{39}$$

$$v_{c,v} = 0.91\frac{d}{l}\sqrt{\frac{\kappa}{C_o}}. \tag{40}$$

Because the maximum angle of rotation is $\alpha_{max} = \frac{d}{l_o}$, the fraction of the angle whereby no pull-in occurs is given as:

$$r_q = \frac{\alpha_q}{\alpha_{max}} = 0.71\frac{l_o}{l} \tag{41}$$

$$r_v = \frac{\alpha_v}{\alpha_{max}} = 0.44\frac{l_o}{l}. \tag{42}$$

Table 4 gives a summary of the above results. With these results, we can now analyze electrostatic pull-in in torsional electrostatic micromirrors.

Table 4
Summary of model equations for electrostatic pull-in in torsional micromirrors under constant charge and constant voltage operation modes.

	Critical angle	Critical voltage	Angle fraction
Constant charge	$\alpha_q = 0.71\frac{d}{l}$	$V_{c,q} = 1.34\frac{d}{l}\sqrt{\frac{\kappa}{C_o}}$	$r_q = \frac{\alpha_q}{\alpha_{max}} = 0.71\frac{l_o}{l}$
Constant voltage	$\alpha_v = 0.44\frac{d}{l}$	$V_{c,v} = 0.91\frac{d}{l}\sqrt{\frac{\kappa}{C_o}}$	$r_v = \frac{\alpha_v}{\alpha_{max}} = 0.44\frac{l_o}{l}$

6.6 *Analysis of electrostatic pull-in*

• Unlike the case of the vertically actuated parallel-plate capacitor, pull-in still occurs in the torsional actuator even under a constant charge operation. The reason is that the torque τ is not necessarily constant even when the charge q is constant due to the nonconstant term $g'(x)$ in eq. (30). However, eqs. (37, 38) and (41, 42) suggest that for a given geometry (l_o, l, d), the onset of pull-in can be delayed under a constant charge operation. Figure 14(a) shows the angle of rotation as a function of applied bias in constant charge and constant voltage operation modes. The critical angle under constant charge operation is larger than that of constant voltage operation, as predicted by eqs. (37, 38).

• In contrast to the vertically actuated parallel-plate capacitor, we can eliminate pull-in in a torsional actuator even under a constant voltage operation by changing its geometry. For both operation modes, this can be achieved by choosing l small enough so that the angle fraction r_q (or r_v) in eq. (41) (or (42)) becomes 1. Figure 14(b) shows how angle fraction r_q can be increased towards 1 by increasing the ratio $\frac{l_o}{l}$. However, this involves a trade-off. A smaller value of l or a larger value of d increases the critical angles and decreases the initial capacitance C_o, resulting in an increase in the required applied voltage for the same angle of rotation.

• Equations (39, 40) provide a method to extract the torsion constant κ. In fact, κ can be expressed in terms of the critical voltages $V_{c,q}$ and the geometry d, l, C_o. It appears that κ should depend neither on l nor d; any relationship with l and d will be absorbed in the critical voltages $V_{c,q}$ and $V_{c,v}$. Here for a given geometry, we can determine d, l, C_o and by measuring the critical voltages $V_{c,q}$ or $V_{c,v}$, we can then extract κ. In the result shown in Fig. 13, we chose $V_c = 12.5$ V, following which we retrieved the value of κ needed for SPICE simulations.

7. Conclusion

We have presented a method for circuit synthesis of coupling elements using multivariate polynomial dependent sources. As illustrated, the method is effective in synthesis of equivalent circuits for electrothermal and electromechanical microsystems. The resulting circuits are simple and can be simulated using standard circuit simulators such as SPICE for system-level simulation, sensitivity analysis, and device optimization. Based on the model equation developed, we investigated electrostatic pull-in in torsional actuators from the point of view of constant charge and constant voltage operation modes along with the necessary model equation for extraction of the torsion constant.

Acknowledgments

This work was supported by the Natural Science and Engineering Research Council of Canada.



(a)



(b)

Fig. 14. (a) A comparison of rotation behavior associated with electrostatic pull-in under constant voltage (solid curve) and constant charge (dashed curve) operation modes. (b) A comparison of rotation behavior under constant charge operation for different micromirror electrod geometries: l_o/l = 20 μm/16 μm (solid curve) and l_o/l = 20 μm/18 μm (dashed curve).

Appendix A
SPICE-code for electro-thermal resistor

```
temperature as a function of input voltage
.OPTION POST=2
.PARAM Alpha=3.7e-3 *  Temperature coefficient
.PARAM Ro=1e4 *10K ohm  Independent resistor
.PARAM InvRo='1/Ro'
.PARAM Rth=1e5 *1M ohm  Thermal resistor for heat convection
.PARAM Cth=1e-4 *100 uF  Thermal capacitor
.PARAM Tambient=300 *300 K  Ambient temperature

** Electric side
 * Input voltage Vin is of pulse form
 * Independent resistor Ro
 * Dependent voltage source (E2): V2=V12*Alpha*T
Vin 1 0 pulse(0 5 50 2 2 100 200 )
Rl  1 2 Ro
E2 2 0 poly(2) 1 2  3 0   0 0 0 0 Alpha

** Thermal side
 * Temperature is the voltage V(3)
 * Dependent current source (G3): I=Vin*V23/Ro
G3  0 3 poly(2) 1 0  1 2   0 0 0 0  InvRo
Cth 3 0 Cth IC=Tambient
Rth 3 4 Rth
Vt  4 0 Tambient

** Simulation starting at 0s and ends at 400s with step 1s
 * Recording input voltage V=v(1), temperature in C=v(3)-273,
.tran 1 400
.print tran vin=v(1) Temp=PAR('v(3)-273')
.end
```

Appendix B
SPICE-code for circuit in Fig. 6 (const. voltage)

```
.OPTION post=2
.PARAM Rin=1e3
.PARAM Co=3.41e-12                    *Co
.PARAM Gap=1e-6                       *d
.PARAM SpringCoeff=2.4049e4           *k
.PARAM SpringCap='1/SpringCoeff'
.PARAM InvKD='-1/(Gap*SpringCoeff)'   *1/kd
.PARAM CoOverGap='Co/(2*Gap)'
.PARAM Mass=1e-6                      *m
.PARAM Damper=1e-1                    *b

** Electrical operation
 * Dependent voltage source E3: V3=-V23*V6/kd (node 3 and 0)
Vin 1 0
```

```
R1 1 2 Rin
C2 2 3 Co IC=0
E3 3 0 poly(2) 2 3 6 0   0 0 0 0  InvKD

** Mechanical operation:
 * Force: Dependent voltage source E4: V4=Co/(2d)*V23^2
E4 4 0  poly(1)  2 3     0 0 CoOverGap
L4 4 5  Mass
R5 5 6  Damper
C6 6 0  SpringCap IC=0

** DC-Analysis
.dc Vin 0 45 1
.print dc Gap= PAR('Gap–V(6)*SpringCap')
.end
```

Appendix C
SPICE-code for circuit in Fig. 6 (const. charge)

```
.OPTION post=2
.PARAM Rin=1e3
.PARAM Co=3.41e–12                    *Co
.PARAM Gap=1e–6                       *d
.PARAM SpringCoeff=2.4049e4           *k
.PARAM SpringCap='1/SpringCoeff'
.PARAM InvKD='–1/(Gap*SpringCoeff)'   *1/kd
.PARAM CoOverGap='Co/(2*Gap)'
.PARAM Mass=1e–6                      *m
.PARAM Damper=1e–1                    *b

** Electrical operation
 * Dependent voltage source E3: V3=–V23*V6/kd (node 3 and 0)
 * Constant charge is simulated by applying a voltage source across Co
R1 0 2 Rin
Vin 2 3
C2 2 3 Co IC=0
E3 3 0 poly(2) 2 3 6 0   0 0 0 0  InvKD

** Mechanical operation:
 * Force: Dependent voltage source E4: V4=Co/(2d)*V23^2
E4 4 0  poly(1)  2 3     0 0 CoOverGap
L4 4 5  Mass
R5 5 6  Damper
C6 6 0  SpringCap IC=0

** DC-Analysis
.dc Vin 0 118 1
.print dc Gap=PAR('Gap-V(6)*SpringCap')
.end
```

References

1 D. Karnopp, D. Margolis and R. Rosenberg: System Dynamics: A Unified Approach (John Wiley, Boston, 2 ed., 1990).
2 A. Vladimirescu: The SPICE Book (John Wiley, Toronto, 1994).
3 H. Pham and A. Nathan: Sensors and Materials **10** (1998) 63.
4 H. Tilmans: J. Micromech. Microeng **6** (1996) 157.
5 N. R. Swart and A. Nathan: Sensors and Materials **6** (1994) 179.
6 W. K. Giloi: Principles of Continuous System Simulation (Teubner Studienbücher Informatik, Stuttgart, 1975).
7 N. M. Karayanakis: Computer-Assisted Simulation of Dynamic Systems with Block Diagram Languages (CRC Press, Boca Raton, Florida, 1993).
8 J. Funk, J. Korvink, M. Bächtold, J. Bühler and H. Baltes: Proc. IEEE MEMS (1996) p. 133.
9 R. Boysel, J. Florence and W. Wu: Proc. SPIE Vol. 1151 (1989) p. 183.
10 S. Senturia: Tranducers'95 — Eurosensors IX (Stockholm, 1995) p. 5.
11 G. K. Fedder and R. T. Howe: J. Microelectromechanical Systems **5** (1996) 283.
12 P. Osterberg, H. Yie, X. Cai, J. White and S. Senturia: Proc. IEEE MEMS (1994) p. 28.
13 R. F. Harrington: Field Computation by Moment Methods (IEEE Press, 1993), Chap. 1.
14 K. Nabors, S. Kims and J. White: IEEE Trans. on Microwave Theory and Techniques. **7** (1992) 1496.
15 H. Pham and A. Nathan: Can. J. of Physics. **75** (1997) 689.
16 H. Pham and A. Nathan: Proc. Royal Society London Series A (to appear).

About the Authors

A. Nathan received his PhD in Electrical Engineering from the University of Alberta, Edmonton, Alberta, Canada, in 1988, where he was engaged in research related to the physics and numerical modeling of semiconductor microsensors.

In 1987, he joined LSI Logic Corp., Santa Clara, CA where he worked on advanced multichip packaging techniques and related issues. Subsequently, he was at the Institute of Quantum Electronics, ETH Zürich, Switzerland. In 1989, he joined the Department of Electrical and Computer Engineering, University of Waterloo, where he is currently a Professor. In 1995, he was a Visiting Professor at the Physical Electronics Laboratory, ETH Zürich. His present research interests lie in large area optical and X-ray imaging using amorphous and crystalline silicon technologies. He currently holds the DALSA/ NSERC industrial research chair in sensor technology.

Hoan H. Pham received his PhD in Electrical Engineering from the University of Waterloo, Waterloo, Ontario, Canada, in 1998 and the BMATH and MMATH degrees in computer science also from the University of Waterloo, in 1994 and 1995, respectively.

His research interest is in development of computer-aided design (CAD) tools for applications in VLSI design, micro-electro-mechanical systems (MEMS), and large-area electronics for displays and digital imaging. He has been actively engaged in SPICE modeling and circuit synthesis of MEMS, and numerical simulation of amorphous silicon (a-Si) thin film transistors (TFTs) for large-area imaging arrays. He is currently involved in the development of new techniques and algorithms for fast capacitance extraction of VLSI interconnects including electrostatic interaction analysis in MEMS.